

graph twoway pcarrowi — Twoway pcarrow with immediate arguments

[Description](#)
[Options](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Also see](#)

[Syntax](#)

Description

`pcarrowi` is an immediate version of `twoway pcarrow`; see [\[U\] 19 Immediate commands](#) and [\[G-2\] graph twoway pcarrow](#). `pcarrowi` is intended for programmer use but can be useful interactively.

Quick start

Directional arrow plot from (5, 1) to (4, 7)

```
twoway pcarrowi 1 5 7 4
```

Overlay arrow plot on a scatterplot of observed values of `x` and `y`

```
twoway scatter y x || pcarrowi 36 22 40 21
```

Same as above, labeling the arrow “My label” at 2 o’clock

```
twoway scatter y x || pcarrowi 36 22 40 21 (2) "My label"
```

Same as above, but specify a large font for the label and place label near the arrowhead

```
twoway scatter y x || pcarrowi 36 22 40 21 (2) "My label", ///
mlabsize(large) headlabel
```

Specify a large arrowhead and a medium-small barb size

```
twoway pcarrowi 36 22 40 21, msize(large) barbsize(medsmall)
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway pcarrowi immediate_values [ , options ]
```

where *immediate_values* is one or more of

```
#y1 #x1 #y2 #x2 [ (#clockposstyle) ] [ "text for label" ]
```

See [G-4] *clockposstyle* for a description of **#clockposstyle**.

Options

options are as defined in [G-2] **graph twoway pcarrow**, with the following modifications:

If "*text for label*" is specified among any of the immediate arguments, option **mlabel()** is assumed.

If (**#clockposstyle**) is specified among any of the immediate arguments, option **mlabvposition()** is assumed.

Remarks and examples

[stata.com](http://www.stata.com)

Immediate commands are commands that obtain data from numbers typed as arguments. Typing

```
. twoway pcarrowi 1.1 1.2 1.3 1.4 2.1 2.2 2.3 2.4, any_options
```


produces the same graph as typing

```
. clear
. input y1 x1 y2 x2
      y1      x1      y2      x2
1. 1.1 1.2 1.3 1.4
2. 2.1 2.2 2.3 2.4
3. end
. twoway pcarrowi y x, any_options
```

twoway pcarrowi does not modify the data in memory.


pcarrowi is intended for programmer use but can be used interactively. In *Basic use* of [G-2] **graph twoway pcarrow**, we drew some simple clock hands from data that we input. We can draw the same graph by using **pcarrowi**.

```
. twoway pcarrowi 0 0 0 1 0 0 1 0
```



We can also draw the annotated second example,

```
. twoway pcarrowi 0 0 0 1 (3) "3 o'clock"
  0 0 1 0 (12) "12 o'clock",
  aspect(1) headlabel plotregion(margin(vlarge))
```




As another example, in [\[G-3\] added_text_options](#), we demonstrated the use of option `text()` to add text to a graph:

```
. twoway qfitci mpg weight, stdf ||
  scatter mpg weight, ms(0)
  text(41 2040 "VW Diesel", place(e))
  text(28 3260 "Plymouth Arrow", place(e))
  text(35 2050 "Datsun 210 and Subaru", place(e))
```

Below we use `pcarrowi` to obtain similar results:

```
. twoway qfitci mpg weight, stdf ||
  scatter mpg weight, ms(0) ||
  pcarrowi 41 2200 41 2060 (3) "VW Diesel"
           28 3460 28 3280 (3) "Plymouth Arrow"
           35 2250 35 2070 (3) "Datsun 210 and Subaru",
  legend(order(1 2 3))
```



Also see

[G-2] **graph twoway** — Twoway graphs

[G-2] **graph twoway pcarrow** — Paired-coordinate plot with arrows

[G-2] **graph twoway scatteri** — Scatter with immediate arguments

[U] **19 Immediate commands**

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).